

EXCEL FORMULA REFERENCE

Excel Formulas Cheat Sheet

Essential Excel formulas, syntax and practical examples for lookups, logic, text, dates, dynamic arrays and errors.

 **cheatsheetsilo.com**
LEARN FASTER. WORK SMARTER.

Excel Formulas Cheat Sheet


Find, learn and use the most useful Excel formulas with clear examples and practical tips.

 **100+**
Formulas

 **Clear**
Examples

 **Practical**
Tips

 **Easy**
Reference



PRINTABLE REFERENCE

22 sections | Practical examples | Quick formula reference

cheatsheetsilo.com

CONTENTS

Excel Formulas Cheat Sheet Contents

01 Excel Formula Quick Reference	02 Formula Basics
03 Cell References	04 Math & Statistics
05 IF & IFS	06 Logical Functions
07 SUMIF & SUMIFS	08 COUNTIF & COUNTIFS
09 XLOOKUP	10 VLOOKUP
11 INDEX & MATCH	12 Text Functions
13 Date & Time	14 Rounding Functions
15 Dynamic Arrays	16 FILTER, SORT & UNIQUE
17 Error Handling	18 Formula Operators
19 Common Formula Errors	20 Practical Examples
21 Best Practices	22 FAQ

01 - EXCEL FORMULA QUICK REFERENCE

Excel Formula Quick Reference

Start here when you need the most common functions at a glance.

Function	Purpose	Example
SUM	Adds values	=SUM(B2:B10)
AVERAGE	Returns the arithmetic mean	=AVERAGE(B2:B10)
COUNT	Counts numeric cells	=COUNT(B2:B100)
IF	Returns one result or another	=IF(B2>=70,"Pass","Fail")
SUMIFS	Sums with multiple criteria	=SUMIFS(C2:C100,A2:A100,"East",B2:B100,"Apples")
COUNTIFS	Counts with multiple criteria	=COUNTIFS(A2:A100,"East",B2:B100,">=100")
XLOOKUP	Looks up and returns a matching value	=XLOOKUP(A2,D2:D100,E2:E100,"Not found")
VLOOKUP	Legacy vertical lookup	=VLOOKUP(A2,D2:G100,4,FALSE)
INDEX + MATCH	Flexible lookup combination	=INDEX(E2:E100,MATCH(A2,D2:D100,0))
TEXTJOIN	Joins text with a delimiter	=TEXTJOIN(", ",TRUE,A2:A5)
TODAY	Returns today's date	=TODAY()
EOMONTH	Returns month-end date	=EOMONTH(A2,0)
FILTER	Returns rows matching criteria	=FILTER(A2:C100,C2:C100>1000,"No results")
UNIQUE	Returns distinct values	=UNIQUE(A2:A100)
IFERROR	Returns fallback on any error	=IFERROR(A2/B2,0)

02 - FORMULA BASICS

Formula Basics

Excel formulas begin with =. They can combine cell references, constants, operators and functions.

Function	Purpose	Example
Arithmetic	Add, subtract, multiply or divide	=A2+B2
Function	Use a built-in function	=SUM(B2:B10)
Precedence	Multiplication before addition	=10+5*2
Parentheses	Control calculation order	=(10+5)*2

TIP: Use parentheses whenever they make the intended calculation easier to read.

03 - CELL REFERENCES

Cell References

Function	Purpose	Example
Relative	Changes when copied	=A2*B2
Absolute	Row and column stay locked	=A2*\$F\$1
Mixed	Locks only row or column	=\$A2*B\$1

SHORTCUT: Press F4 while editing a reference to cycle through relative, absolute and mixed reference styles.

04 - MATH & STATISTICS

Math & Statistics

Function	Purpose	Example
SUM	Add values	=SUM(B2:B100)
AVERAGE	Mean value	=AVERAGE(B2:B100)
COUNT	Count numbers	=COUNT(B2:B100)
COUNTA	Count nonblank cells	=COUNTA(B2:B100)
COUNTBLANK	Count blank cells	=COUNTBLANK(B2:B100)
MIN / MAX	Smallest / largest value	=MAX(B2:B100)
MEDIAN	Middle value	=MEDIAN(B2:B100)
LARGE / SMALL	Nth largest / smallest	=LARGE(B2:B100,3)
PRODUCT	Multiply values	=PRODUCT(B2:B5)
ABS	Absolute value	=ABS(B2)

05 - IF & IFS

IF & IFS

Function	Purpose	Example
IF	Single logical branch	=IF(B2>=70,"Pass","Fail")
IF	Blank check	=IF(A2="", "Missing", "Complete")
IF	Commission	=IF(B2>=1000, B2*10%, 0)

Function	Purpose	Example
Nested IF	Multiple levels	=IF(B2>=90,"A",IF(B2>=80,"B",IF(B2>=70,"C","F")))
IFS	Multiple ordered tests	=IFS(B2>=90,"A",B2>=80,"B",B2>=70,"C",TRUE,"F")

06 - LOGICAL FUNCTIONS

Logical Functions

Function	Purpose	Example
AND	TRUE only when all tests are TRUE	=AND(B2>=70,C2>=80%)
OR	TRUE when any test is TRUE	=OR(B2="Approved",B2="Pending")
NOT	Reverses TRUE/FALSE	=NOT(B2="Cancelled")
IF + AND	Combine tests	=IF(AND(B2>=70,C2>=80%),"Pass","Fail")
IF + OR	Accept either condition	=IF(OR(B2="Approved",B2="Pending"),"Continue","Stop")

07 - SUMIF & SUMIFS

SUMIF & SUMIFS

Function	Purpose	Example
SUMIF	One condition	=SUMIF(A2:A100,"East",C2:C100)
SUMIFS	Multiple conditions	=SUMIFS(C2:C100,A2:A100,"East",B2:B100,"Apples")
SUMIFS	Criteria from cells	=SUMIFS(C2:C100,A2:A100,E2,B2:B100,F2)
SUMIFS	Date range	=SUMIFS(C2:C100,A2:A100,">="&E2,A2:A100,"<="&F2)

08 - COUNTIF & COUNTIFS

COUNTIF & COUNTIFS

Function	Purpose	Example
COUNTIF	One condition	=COUNTIF(A2:A100,"East")
COUNTIF	Numeric condition	=COUNTIF(B2:B100,">=100")
COUNTIFS	Multiple conditions	=COUNTIFS(A2:A100,"East",B2:B100,">=100")
COUNTIF	Wildcard text	=COUNTIF(A2:A100,"*apple*")

09 - XLOOKUP

XLOOKUP

XLOOKUP is the preferred modern lookup for many Excel tasks because it can return values to the left or right and supports a built-in not-found result.

Function	Purpose	Example
Exact lookup	Return matching value	=XLOOKUP(A2,D2:D100,E2:E100,"Not found")
Reverse direction	Return from a column to the left	=XLOOKUP(A2,E2:E100,D2:D100,"Not found")
Approximate	Next smaller match	=XLOOKUP(A2,D2:D100,E2:E100,"Not found",-1)

10 - VLOOKUP

VLOOKUP

Function	Purpose	Example
Exact	Use FALSE for exact match	=VLOOKUP(A2,D2:G100,4,FALSE)
Approximate	Use TRUE for approximate match	=VLOOKUP(A2,D2:E10,2,TRUE)

CAUTION: Approximate VLOOKUP normally expects the first lookup column to be sorted ascending. For new workbooks, XLOOKUP is often easier and safer.

11 - INDEX & MATCH

INDEX & MATCH

Function	Purpose	Example
INDEX	Return value by position	=INDEX(E2:E100,5)
MATCH	Find position	=MATCH(A2,D2:D100,0)
INDEX + MATCH	Flexible exact lookup	=INDEX(E2:E100,MATCH(A2,D2:D100,0))
Two-way lookup	Match row and column	=INDEX(B2:F20,MATCH(H2,A2:A20,0),MATCH(H3,B1:F1,0))

12 - TEXT FUNCTIONS

Text Functions

Function	Purpose	Example
LEFT	Characters from left	=LEFT(A2,3)
RIGHT	Characters from right	=RIGHT(A2,4)
MID	Characters from middle	=MID(A2,3,5)
LEN	Text length	=LEN(A2)
TRIM	Remove extra spaces	=TRIM(A2)
UPPER / LOWER	Change case	=UPPER(A2)
PROPER	Capitalize words	=PROPER(A2)
CONCAT	Join text	=CONCAT(A2,B2)
TEXTJOIN	Join with delimiter	=TEXTJOIN(", ",TRUE,A2:A5)
FIND / SEARCH	Find text position	=SEARCH("excel",A2)
SUBSTITUTE	Replace matching text	=SUBSTITUTE(A2,"old","new")
REPLACE	Replace by position	=REPLACE(A2,1,3,"New")
TEXT	Format a value as text	=TEXT(A2,"mmm d, yyyy")

13 - DATE & TIME

Date & Time

Function	Purpose	Example
TODAY	Current date	=TODAY()
NOW	Current date and time	=NOW()
DATE	Build a date	=DATE(2026,9,5)
YEAR / MONTH / DAY	Extract date parts	=YEAR(A2)

Function	Purpose	Example
WEEKDAY	Day-of-week number	=WEEKDAY(A2,2)
EDATE	Shift by months	=EDATE(A2,3)
EOMONTH	Month-end date	=EOMONTH(A2,0)
NETWORKDAYS	Working days between dates	=NETWORKDAYS(A2,B2)
WORKDAY	Date after working days	=WORKDAY(A2,10)
DATEDIF	Completed years between dates	=DATEDIF(A2,TODAY(),"Y")

NOTE: NETWORKDAYS and WORKDAY can accept a holiday range as an additional argument. DATEDIF remains supported for compatibility but is a legacy function with documented limitations.

14 - ROUNDING FUNCTIONS

Rounding Functions

Function	Purpose	Example
ROUND	Round to digits	=ROUND(A2,2)
ROUNDUP	Always round away from zero	=ROUNDUP(A2,0)
ROUNDDOWN	Always round toward zero	=ROUNDDOWN(A2,0)
INT	Round down to integer	=INT(A2)
MROUND	Nearest multiple	=MROUND(A2,5)
CEILING.MATH	Round up to multiple	=CEILING.MATH(A2,5)
FLOOR.MATH	Round down to multiple	=FLOOR.MATH(A2,5)

NEGATIVE NUMBERS: INT(-3.2) returns -4, while ROUNDDOWN(-3.2,0) returns -3. They are not equivalent for negative values.

15 - DYNAMIC ARRAYS

Dynamic Arrays

Modern Excel can return multiple results from one formula. The results spill into neighboring cells automatically.

Function	Purpose	Example
SEQUENCE	Generate a sequence	=SEQUENCE(10)
UNIQUE	Distinct values	=UNIQUE(A2:A100)
SORT	Sort an array	=SORT(A2:C100,3,-1)
FILTER	Filter rows	=FILTER(A2:C100,C2:C100>1000)
RANDARRAY	Random array	=RANDARRAY(5,3,1,100,TRUE)
Spill operator	Refer to a spilled range	=SUM(B2#)

#SPILL!: A dynamic-array formula cannot spill if cells in its intended output range are blocked.

16 - FILTER, SORT & UNIQUE

FILTER, SORT & UNIQUE

Function	Purpose	Example
FILTER	Single condition	=FILTER(A2:C100,C2:C100>1000,"No results")

Function	Purpose	Example
FILTER AND	Multiply conditions	=FILTER(A2:C100, (B2:B100="East")*(C2:C100>1000), "No results")
FILTER OR	Add conditions	=FILTER(A2:C100, (B2:B100="East")+(B2:B100="West"), "No results")
SORT	Sort by third column descending	=SORT(A2:C100,3, -1)
UNIQUE	Distinct values	=UNIQUE(A2:A100)
Combined	Filter then sort	=SORT(FILTER(A2:C100,C2:C100>1000,"No results"),3,-1)

17 - ERROR HANDLING

Error Handling

Function	Purpose	Example
IFERROR	Fallback for any error	=IFERROR(A2/B2,0)
IFNA	Fallback only for #N/A	=IFNA(XLOOKUP(A2,D2:D100,E2:E100),"Not found")
ISERROR	Test for any error	=ISERROR(A2/B2)
ISNA	Test for #N/A	=ISNA(MATCH(A2,D2:D100,0))

BEST PRACTICE: Do not hide every error automatically. When possible, test the expected condition directly so genuine problems remain visible.

18 - FORMULA OPERATORS

Formula Operators

Operator	Meaning	Example
+ - * /	Arithmetic	=A2*B2
^	Exponent	=A2^2
= > < >= <= <>	Comparison	=A2>=100
&	Text concatenation	=A2&" "&B2
:	Range	=SUM(A2:A10)
space	Intersection	Reference intersection
,	Union in reference contexts	Union behavior depends on formula context and locale

REGIONAL SETTINGS: Examples in this PDF use commas as function argument separators. Some Excel locales use semicolons instead.

19 - COMMON FORMULA ERRORS

Common Formula Errors

Function	Purpose	Example
#DIV/0!	Division by zero or blank divisor	=A2/B2
#N/A	Lookup value not available	=XLOOKUP(A2,D2:D100,E2:E100)
#VALUE!	Wrong value or data type	Check referenced cells
#REF!	Invalid or deleted reference	Check deleted rows/columns/sheets

Function	Purpose	Example
#NAME?	Unrecognized name/function or missing quotes	Check spelling and text quotes
#NUM!	Invalid numeric calculation	=SQRT(-1)
#SPILL!	Blocked dynamic-array output	=UNIQUE(A2:A100)
#####	Usually display width/formatting	Widen the column or check formatting

QUICK FIXES: Use =IF(B2=0,0,A2/B2) for expected zero divisors, or XLOOKUP's built-in "not found" argument for missing lookups.

20 - PRACTICAL EXAMPLES

Practical Examples

Function	Purpose	Example
Sales total	Quantity x price	=B2*C2
Commission	10% above threshold	=IF(B2>=1000,B2*10%,0)
Target status	Compare actual to target	=IF(B2>=C2,"Target Met","Below Target")
Price lookup	Find product price	=XLOOKUP(A2,D2:D100,E2:E100,"Not found")
Due date	Add 30 days	=A2+30
Clean name	Trim and proper case	=PROPER(TRIM(A2))
Regional sales	Conditional sum	=SUMIF(A2:A100,"East",C2:C100)
Workflow	Multiple conditions	=IF(AND(B2>=70,C2="Complete"),"Pass","Review")
Next work date	Add working days	=WORKDAY(A2,10)
Dynamic report	Filter and sort	=SORT(FILTER(A2:C100,C2:C100>1000,"No results"),3,-1)

21 - BEST PRACTICES

Best Practices

01 Use references deliberately
Choose relative, absolute or mixed references based on how the formula will be copied.

02 Avoid hard-coded assumptions
Store rates, targets and editable values in cells instead of burying them in formulas.

03 Keep formulas readable
Break complex logic into understandable steps when needed.

04 Use Tables for growing data
Structured references can make expanding datasets easier to maintain.

05 Use named ranges carefully
Clear names can improve readability for important inputs and ranges.

06 Handle errors intentionally
Use IFERROR or IFNA when the error condition is expected.

07 Test edge cases
Check blanks, zeros, negatives, duplicates and missing lookups.

08 Prefer modern functions when appropriate
XLOOKUP, FILTER, SORT and UNIQUE can simplify many modern Excel workflows.

AUDIT CHECKLIST: Check references, inputs, boundaries, missing data, copied formulas and dependencies before relying on a workbook.

22 - FAQ

FAQ

What is the difference between a formula and a function?

A formula is any Excel calculation beginning with =. A function is a built-in calculation, such as SUM or XLOOKUP, that can be used inside a formula.

How do I lock a cell reference?

Use \$ signs: \$A\$1 locks row and column, \$A1 locks the column, and A\$1 locks the row. F4 cycles reference styles while editing.

Should I use XLOOKUP or VLOOKUP?

For modern Excel, XLOOKUP is usually more flexible. VLOOKUP remains useful for compatibility with older workbooks.

Why is my formula showing as text?

Check whether the cell is formatted as Text, whether the formula begins with an apostrophe, or whether Show Formulas is enabled.

What does #SPILL! mean?

A dynamic-array result cannot expand because cells in the spill range are blocked or another restriction prevents spilling.

Why does my Excel use semicolons instead of commas?

Argument separators depend on regional settings. Some locales use semicolons where these examples use commas.

Why does a lookup return #N/A even when values look identical?

Check extra spaces, text-versus-number data types, hidden characters and whether the lookup value actually exists.

Do formulas update automatically?

Normally yes when calculation mode is Automatic. In Manual calculation mode, recalculation must be triggered.

REFERENCE COMPLETE: Keep this PDF nearby for quick formula syntax and examples. For full explanations and interactive copy buttons, visit [CheatSheetSilo.com](https://cheatsheet.silo.com).

CheatSheetSilo.com - Learn faster. Work smarter.